# Maintainer: Jaroslav Lichtblau <svetlemodry@archlinux.org>
# Contributor: Kaushal M <kshlmster@gmail.com>

pkgname=perl-file-finder
pkgver=0.53
pkgrel=8
pkgdesc="Perl/CPAN Module File::Finder"
arch=('any')
url='https://search.cpan.org/dist/File-Finder/'
license=('GPL' 'PerlArtistic')
depends=('perl')
options=('!emptydirs')
source=(https://search.cpan.org/CPAN/authors/id/M/ME/MERLYN/File-Finder-$pkgver.tar.gz)
sha256sums=('2ecbc19ac67a9e635c872a807a8d3eaaff5babc054f15a191d47cdfc5f176a74')

build() {
  cd "${srcdir}"/File-Finder-$pkgver

  PERL_MM_USE_DEFAULT=1 perl Makefile.PL installdirs=vendor
  make
}

package() {
  cd "${srcdir}"/File-Finder-$pkgver

  make install DESTDIR="${pkgdir}"

  find "${pkgdir}" -name '.packlist' -delete
  find "${pkgdir}" -name '*.pod' -delete
}
